

Example 6: General Linear Models with Single-Slope Interactions (i.e., of Quantitative and Binary) Predictors in SAS and STATA

This example comes from Hoffman (2015) chapter 2, which examined prediction of cognition (as measured by an information test outcome) from age (centered at 85 years) grip strength (centered at 9 pounds), sex (with men as the reference group) and subsequent dementia status (none = 1, future = 2, and current = 3, with none as the reference) in a sample of 550 older adults. Building on the combined final main effects only model of Example 5b, this example begins by showing two ways of generating predicted outcomes using a main effects only model. It then illustrates how to include and interpret interactions: first with sex by age and sex by grip strength, and then with age by grip strength.

SAS Syntax for Importing and Preparing Data for Analysis:

```
* Defining global variable for file location to be replaced in code below;
* \\Client\ precedes path in Virtual Desktop outside H drive;
%LET filesave= C:\Dropbox\21SP_PSQF6242\PSQF6242_Example6;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.";

* Import chapter 2 example data into work library as Example6;
DATA work.Example6; SET filesave.SAS_Chapter2;
* Center quantitative predictors;
age85 = age - 85;
grip9 = grip - 9;
* Create dummy-coded binary predictors for dementia groups;
demNF=.; demNC=.; * Create two new empty variables;
IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * Replace each for none group;
IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Replace each for future group;
IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Replace each for current group;
* Label new variables - note semi-colon is only at the end of ALL labels;
LABEL
age85=      "age85: Age in Years (0=85)"
grip9=      "grip9: Grip Strength in Pounds (0=9)"
sexMW=      "sexMW: Sex (0=M, 1=W)"
demNF=      "demNF: Dementia Contrast for None=0 vs Future=1"
demNC=      "demNC: Dementia Contrast for None=0 vs Current=1"
cognition=  "cognition: Cognition Outcome"
demgroup=   "demgroup: Dementia Group 1N 2F 3C";
* Select cases complete on variables;
IF NMISS(cognition,age,grip,sexmw,demgroup)>0 THEN DELETE;
RUN;
```

STATA Syntax for Importing and Preparing Data for Analysis:

```
// Defining global variable for file location to be replaced in code below
// \\Client\ precedes path in Virtual Desktop outside H drive
global filesave "C:\Dropbox\21SP_PSQF6242\PSQF6242_Example6"

// Import chapter 2 data into temporary file and center predictors
use "$filesave\STATA_Chapter2.dta", clear // Has converted all variables to lower-case

// Center quantitative predictors
gen age85 = age - 85
gen grip9 = grip - 9
// Create age*grip interaction for semi-partial eta2
gen agegrip = age85*grip9
// Create dummy-coded binary predictors for dementia groups
gen demnf=.
gen demnc=.
// Demgroup = none
replace demnf=0 if demgroup==1
replace demnc=0 if demgroup==1
```

```
// Demgroup = future
replace demnf=1 if demgroup==2
replace demnc=0 if demgroup==2
// Demgroup = current
replace demnf=0 if demgroup==3
replace demnc=1 if demgroup==3
// Label all variables
label variable age85      "age85: Age in Years (0=85)"
label variable grip9      "grip9: Grip Strength in Pounds (0=9)"
label variable sexmw      "sexmw: Sex (0=Men, 1=Women)"
label variable demnf      "demnf: Dementia Contrast for None=0 vs Future=1"
label variable demnc      "demnc: Dementia Contrast for None=0 vs Current=1"
label variable cognition   "cognition: Cognition Outcome"
label variable demgroup   "demgroup: Dementia Group 1N 2F 3C"
// Select cases complete on variables
egen nmiss=rowmiss(cognition age grip sexmw demgroup)
drop if nmiss>0
```

SAS Syntax and Output with Main Effects Only of All Predictors of Cognition: Demonstrating how to get predicted outcomes using ESTIMATE statements and plot them

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + e_i$$

```
TITLE1 "SAS Combined Main Effects Only Model Predicting Cognition";
TITLE2 "Demonstrating how to get predicted outcomes using ESTIMATE statements";
PROC GLM DATA=work.Example6 NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
* We are ignoring the effects we would normally request for dementia for now;
* Pred cognition outcomes holding sexMW=men, demNF=none, and demNC=none;
ESTIMATE "Yhat for Age=80 Grip=6" intercept 1 age85 -5 grip9 -3 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=80 Grip=9" intercept 1 age85 -5 grip9 0 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=80 Grip=12" intercept 1 age85 -5 grip9 3 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=85 Grip=6" intercept 1 age85 0 grip9 -3 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=85 Grip=9" intercept 1 age85 0 grip9 0 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=85 Grip=12" intercept 1 age85 0 grip9 3 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=90 Grip=6" intercept 1 age85 5 grip9 -3 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=90 Grip=9" intercept 1 age85 5 grip9 0 sexMW 0 demNF 0 demNC 0;
ESTIMATE "Yhat for Age=90 Grip=12" intercept 1 age85 5 grip9 3 sexMW 0 demNF 0 demNC 0;
ODS OUTPUT Estimates=work.EstMainEffects; * Save ESTIMATEs to dataset for plotting;
RUN; QUIT; TITLE1; TITLE2;
```

SAS Combined Main Effects Only Model Predicting Cognition

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	18385.97930	3677.19586	41.75	<.0001
Error	544	47910.55888	88.07088		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.277329	37.80790	9.384609	24.82182

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits		
Intercept	29.26432541	0.69850792	41.90	<.0001	27.89222232	30.63642850	Beta0
age85	-0.40573396	0.11889717	-3.41	0.0007	-0.63928775	-0.17218017	Beta1
grip9	0.60422556	0.14977568	4.03	<.0001	0.31001605	0.89843507	Beta2
sexMW	-3.65737421	0.89143262	-4.10	<.0001	-5.40844590	-1.90630252	Beta3
demNF	-5.72197100	1.01907848	-5.61	<.0001	-7.72378184	-3.72016016	Beta4
demNC	-16.47981327	1.52275357	-10.82	<.0001	-19.47101037	-13.48861616	Beta5

Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits	
Yhat for Age=80 Grip=6	29.4803185	1.15590606	25.50	<.0001	27.2097326	31.7509045
Yhat for Age=80 Grip=9	31.2929952	0.92090860	33.98	<.0001	29.4840228	33.1019676
Yhat for Age=80 Grip=12	33.1056719	0.87396571	37.88	<.0001	31.3889110	34.8224327
Yhat for Age=85 Grip=6	27.4516487	0.93731216	29.29	<.0001	25.6104543	29.2928432
Yhat for Age=85 Grip=9	29.2643254	0.69850792	41.90	<.0001	27.8922223	30.6364285
Yhat for Age=85 Grip=12	31.0770021	0.70785742	43.90	<.0001	29.6865335	32.4674707
Yhat for Age=90 Grip=6	25.4229789	1.06198691	23.94	<.0001	23.3368816	27.5090763
Yhat for Age=90 Grip=9	27.2356556	0.91355395	29.81	<.0001	25.4411302	29.0301810
Yhat for Age=90 Grip=12	29.0483323	0.97218055	29.88	<.0001	27.1386447	30.9580199

```

* Labeling saved ESTIMATES for use in plot;
* INDEX finds value in parentheses for that column;
DATA work.EstMainEffects; SET work.EstMainEffects;
  IF INDEX(Parameter,"Age=80")>0 THEN age=80;
  IF INDEX(Parameter,"Age=85")>0 THEN age=85;
  IF INDEX(Parameter,"Age=90")>0 THEN age=90;
  IF INDEX(Parameter,"Grip=6")>0 THEN grip=6;
  IF INDEX(Parameter,"Grip=9")>0 THEN grip=9;
  IF INDEX(Parameter,"Grip=12")>0 THEN grip=12; RUN;

* Plot ESTIMATES -- grip as X by age;
PROC SGPLOT DATA=work.EstMainEffects;
  SERIES x=grip y=Estimate / GROUP=age;
  XAXIS GRID LABEL="Grip Strength" VALUES=(6 TO 12 BY 3);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5); RUN; QUIT;

```

Demonstrating how to get predicted outcomes using “fake people” and plot them


```

* Demonstrating how to get predicted outcomes using "fake people";
* Each row is a fake person for which to create a predicted outcome;
DATA work.FakePeople; * List variables;
INPUT PersonID age grip sexMW demNF demNC;
* Center predictors;
  age85=age-85; grip9=grip-9;
* Enter new data;
  DATALINES;
-99 80 6 0 0 0
-99 80 9 0 0 0
-99 80 12 0 0 0
-99 85 6 0 0 0
-99 85 9 0 0 0
-99 85 12 0 0 0
-99 90 6 0 0 0
-99 90 9 0 0 0
-99 90 12 0 0 0
; RUN;
* Merge with real data;
DATA work.Example6;
SET work.FakePeople work.Example6; RUN;

TITLE1 "SAS Combined Main Effects Only Model Predicting Cognition";
TITLE2 "Using dataset with fake people to get predicted outcomes as saved variable";
PROC GLM DATA=work.Example6 NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
* We are ignoring the extra effects we would normally request for dementia for now;
* Request columns of predicted outcome and SE for all cases;
  OUTPUT OUT=work.PredOutcomes PREDICTED=Yhat STDP=SEyhat; RUN; QUIT; TITLE1; TITLE2;

* Plot saved predicted values for fake people -- grip as X by age;
PROC SGPLOT DATA=work.PredOutcomes; WHERE PersonID=-99; * Only for fake people;
  SERIES x=grip y=Yhat / GROUP=age;
  XAXIS GRID LABEL="Grip Strength" VALUES=(6 TO 12 BY 3);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5); RUN; QUIT;

```



**STATA Syntax and Output with Main Effects Only of All Predictors of Cognition:
Demonstrating how to get predicted outcomes using MARGINS statement and plot them**

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + e_i$$

```
display "STATA Combined Main Effects Only Model Predicting Cognition"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc, level(95)
// We are ignoring the effects we would normally request for dementia for now
// Pred cognition outcomes holding sexMW=men, demNF=none, and demNC=none
// one margins replaces 9 ESTIMATES in SAS; vsquish compresses output empty lines
margins, at(c.age85=(-5(5)5) c.grip9=(-3(3)3) c.sexmw=0 c.demnf=0 c.demnc=0) vsquish
// Get and save plot of predicted outcomes and save
marginsplot, xdimension(grip9) name(predicted_means, replace)
```

Source	SS	df	MS	Number of obs	=	550
Model	18385.9793	5	3677.19586	F(5, 544)	=	41.75
Residual	47910.5589	544	88.0708803	Prob > F	=	0.0000
				R-squared	=	0.2773
				Adj R-squared	=	0.2707
Total	66296.5382	549	120.758722	Root MSE	=	9.3846

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
age85	-.405734	.1188972	-3.41	0.001	-.6392878	-.1721802
grip9	.6042256	.1497757	4.03	0.000	.310016	.8984351
sexmw	-3.657374	.8914326	-4.10	0.000	-5.408446	-1.906303
demnf	-5.721971	1.019078	-5.61	0.000	-7.723782	-3.72016
demnc	-16.47981	1.522754	-10.82	0.000	-19.47101	-13.48862
_cons	29.26433	.6985079	41.90	0.000	27.89222	30.63643

Beta1

Beta2

Beta3


Beta4

Beta5

Beta0

		Margin	Delta-method Std. Err.	t	P> t	[95% Conf. Interval]	
_at	1	29.48032	1.155906	25.50	0.000	27.20973	31.7509
	2	31.293	.9209086	33.98	0.000	29.48402	33.10197
	3	33.10567	.8739657	37.88	0.000	31.38891	34.82243
	4	27.45165	.9373122	29.29	0.000	25.61045	29.29284
	5	29.26433	.6985079	41.90	0.000	27.89222	30.63643
	6	31.077	.7078574	43.90	0.000	29.68653	32.46747
	7	25.42298	1.061987	23.94	0.000	23.33688	27.50908
	8	27.23566	.913554	29.81	0.000	25.44113	29.03018
	9	29.04833	.9721806	29.88	0.000	27.13864	30.95802

Although annoying that they are not labeled here, a long table preceded this MARGINS result that says what the predictor values are for each of these 9 predicted outcomes.



FOR HW6 → SAS Syntax adding Two Interactions: Sex by Age and Sex by Grip Strength

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{SexMW}_i)(\text{Age}_i - 85) + \beta_7(\text{SexMW}_i)(\text{Grip}_i - 9) + e_i$$

```
TITLE1 "SAS GLM adding 2 Interactions (Sex by Age, Sex by Grip)";
PROC GLM DATA=work.Example6 NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC
               sexMW*age85 sexMW*grip9 / ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
ESTIMATE "Mean Diff: Fut vs. Cur"      demNF -1 demNC 1; * B5-B4;
CONTRAST "DFnum=2 F-test for Demgroup" demNF 1, demNC 1;
CONTRAST "DFnum=2 F-test for new interactions" sexMW*age85 1, sexMW*grip9 1;
```

We can use the model equation to calculate the **simple age slope** for either sex (as the moderator):

$$\begin{aligned} \text{Simple Age Slope} &= \beta_1(\text{Age}_i - 85) + \beta_6(\text{SexMW}_i)(\text{Age}_i - 85) \\ &= [\beta_1 + \beta_6(\text{SexMW}_i)] \text{ that multiplies } (\text{Age}_i - 85) \end{aligned}$$

```
* Simple slopes of age by sex;
ESTIMATE "Age Slope for Men"      age85 1 sexMW*age85 0;
ESTIMATE "Age Slope for Women"    age85 1 sexMW*age85 1;
```

We can use the model equation to calculate the **simple grip slope** for either sex (as the moderator):

$$\begin{aligned} \text{Simple Grip Slope} &= \beta_2(\text{Grip}_i - 9) + \beta_7(\text{SexMW}_i)(\text{Grip}_i - 9) \\ &= [\beta_2 + \beta_7(\text{SexMW}_i)] \text{ that multiplies } (\text{Grip}_i - 9) \end{aligned}$$

```
* Simple slopes of grip by sex;
ESTIMATE "Grip Slope for Men"      grip9 1 sexMW*grip9 0;
ESTIMATE "Grip Slope for Women"    grip9 1 sexMW*grip9 1;

* If you are NOT using fake people, you have to write these to create predicted outcomes;
* Pred cognition outcomes holding demNF=none and demNC=none;
ESTIMATE "Yhat Men Age=80 Grip=6"  intercept 1 sexMW 0 age85 -5 grip9 -3 sexMW*age85 0 sexMW*grip9 0;
ESTIMATE "Yhat Men Age=80 Grip=12" intercept 1 sexMW 0 age85 -5 grip9 3 sexMW*age85 0 sexMW*grip9 0;
ESTIMATE "Yhat Men Age=90 Grip=6"  intercept 1 sexMW 0 age85 5 grip9 -3 sexMW*age85 0 sexMW*grip9 0;
ESTIMATE "Yhat Men Age=90 Grip=12" intercept 1 sexMW 0 age85 5 grip9 3 sexMW*age85 0 sexMW*grip9 0;
ESTIMATE "Yhat Women Age=80 Grip=6" intercept 1 sexMW 1 age85 -5 grip9 -3 sexMW*age85 -5 sexMW*grip9 -3;
ESTIMATE "Yhat Women Age=80 Grip=12" intercept 1 sexMW 1 age85 -5 grip9 3 sexMW*age85 -5 sexMW*grip9 3;
ESTIMATE "Yhat Women Age=90 Grip=6" intercept 1 sexMW 1 age85 5 grip9 -3 sexMW*age85 5 sexMW*grip9 -3;
ESTIMATE "Yhat Women Age=90 Grip=12" intercept 1 sexMW 1 age85 5 grip9 3 sexMW*age85 5 sexMW*grip9 3;
ODS OUTPUT Estimates=EstSexInteract; * Save estimates to dataset;
RUN; QUIT; TITLE1; TITLE2;
```

FOR HW6 → STATA Syntax adding Two Interactions: Sex by Age and Sex by Grip Strength

```
display "STATA GLM adding 2 Interactions (Sex by Age and Sex by Grip)"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc /// line continuer
               c.sexmw#c.age85 c.sexmw#c.grip9, level(95)
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
test (c.demnf=0) (c.demnc=0) // DFnum=2 F-test for Demgroup
test (c.sexmw#c.age85=0) (c.sexmw#c.grip9=0) // DFnum=2 F-test for two new interactions
```

We can use the model equation to calculate the **simple age slope** for either sex (as the moderator):

$$\begin{aligned} \text{Simple Age Slope} &= \beta_1(\text{Age}_i - 85) + \beta_6(\text{SexMW}_i)(\text{Age}_i - 85) \\ &= [\beta_1 + \beta_6(\text{SexMW}_i)] \text{ that multiplies } (\text{Age}_i - 85) \end{aligned}$$

```
// Simple slopes of age by sex
lincom c.age85*1 + c.sexmw#c.age85*0 // Age Slope for Men
lincom c.age85*1 + c.sexmw#c.age85*1 // Age Slope for Women
```

We can use the model equation to calculate the **simple grip slope** for either sex (as the moderator):

$$\begin{aligned}\text{Simple Grip Slope} &= \beta_2(\text{Grip}_i - 9) + \beta_7(\text{SexMW}_i)(\text{Grip}_i - 9) \\ &= [\beta_2 + \beta_7(\text{SexMW}_i)] \text{ that multiplies } (\text{Grip}_i - 9)\end{aligned}$$

```
// Simple slopes of grip by sex
lincom c.grip9*1 + c.sexmw#c.grip9*0 // Grip Slope for Men
lincom c.grip9*1 + c.sexmw#c.grip9*1 // Grip Slope for Women

// one margins replaces 8 ESTIMATES in SAS; vsquish compresses output empty lines
// predictor=(from(by)to), c.=quantitative predictor
margins, at(c.age85=(-5(10)5) c.grip9=(-3(6)3) c.sexmw=(0(1)1) c.demnf=0 c.demnc=0) vsquish
marginsplot, xdimension(age85) bydimension(grip9) // Plot pred outcomes by age
marginsplot, xdimension(grip9) bydimension(age85) // Plot pred outcomes by grip
```

SAS Model Output:

SAS GLM with 2 Interactions (Sex by Age, Sex by Grip)

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	7	18529.39696	2647.05671	30.04	<.0001
Error	542	47767.14122	88.13126		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.279493	37.82086	9.387825	24.82182

Source	Total Variation Accounted For				
	Semipartial Eta-Square	Semipartial	Conservative		
		Omega-Square	95% Confidence Limits		
age85	0.0020	0.0007	0.0000	0.0162	ok, but conditional
grip9	0.0113	0.0100	0.0005	0.0350	ok, but conditional
sexMW	0.0211	0.0197	0.0039	0.0504	ok, but conditional
demNF	0.0422	0.0408	0.0154	0.0795	not ok, needs combined
demNC	0.1559	0.1543	0.1044	0.2103	not ok, needs combined
age85*sexMW	0.0019	0.0006	0.0000	0.0160	ok, unconditional
grip9*sexMW	0.0004	-0.0009	0.0000	0.0106	ok, unconditional
DFnum=2 F-test for Demgroup	0.1786	0.1757	0.1231	0.2330	ok, from CONTRAST
DFnum=2 F-test for new interactions	0.0022	-0.0005	0.0000	0.0136	ok, from CONTRAST

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
DFnum=2 F-test for Demgroup	2	11843.12235	5921.56117	67.19	<.0001
DFnum=2 F-test for new interactions	2	143.41767	71.70883	0.81	0.4438

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits		
Intercept	29.17207481	0.75835647	38.47	<.0001	27.68239690	30.66175272	Beta0
age85	-0.23224895	0.18939121	-1.23	0.2206	-0.60427967	0.13978178	Beta1
grip9	0.69688281	0.23879978	2.92	0.0037	0.22779635	1.16596927	Beta2
sexMW	-3.62653419	0.91114541	-3.98	<.0001	-5.41634312	-1.83672526	Beta3
demNF	-5.74882607	1.02015364	-5.64	<.0001	-7.75276537	-3.74488677	Beta4
demNC	-16.49475985	1.52333719	-10.83	<.0001	-19.48712801	-13.50239170	Beta5
age85*sexMW	-0.29478557	0.24394817	-1.21	0.2274	-0.77398527	0.18441413	Beta6
grip9*sexMW	-0.17672735	0.30663085	-0.58	0.5646	-0.77905782	0.42560311	Beta7

Interpret these fixed effects:Simple main effect of Age $\beta_1 =$ Simple main effect of Grip Strength $\beta_2 =$ Interpret Sex by Age $\beta_6 \rightarrow$ Age as Simple Slope, Sex as Moderator:Interpret Sex by Grip Strength $\beta_7 \rightarrow$ Grip as Simple Slope, Sex as Moderator:**Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits	
Mean Diff: Fut vs. Cur	-10.7459338	1.70875193	-6.29	<.0001	-14.1025215	-7.3893461
Age Slope for Men	-0.2322489	0.18939121	-1.23	0.2206	-0.6042797	0.1397818
Age Slope for Women	-0.5270345	0.15376644	-3.43	0.0007	-0.8290857	-0.2249833
Grip Slope for Men	0.6968828	0.23879978	2.92	0.0037	0.2277964	1.1659693
Grip Slope for Women	0.5201555	0.19305890	2.69	0.0073	0.1409201	0.8993908
Yhat: Men Age=80 Grip=6	28.2426711	1.59490110	17.71	<.0001	25.1097264	31.3756159
Yhat: Men Age=80 Grip=12	32.4239680	1.14540612	28.31	<.0001	30.1739889	34.6739471
Yhat: Men Age=90 Grip=6	25.9201816	1.56654423	16.55	<.0001	22.8429397	28.9974236
Yhat: Men Age=90 Grip=12	30.1014785	1.27691341	23.57	<.0001	27.5931730	32.6097840
Yhat: Women Age=80 Grip=6	26.6202468	1.11819135	23.81	<.0001	24.4237271	28.8167666
Yhat: Women Age=80 Grip=12	29.7411796	1.11582861	26.65	<.0001	27.5493011	31.9330581
Yhat: Women Age=90 Grip=6	21.3499017	0.95706096	22.31	<.0001	19.4698985	23.2299048
Yhat: Women Age=90 Grip=12	24.4708344	1.33798219	18.29	<.0001	21.8425684	27.0991004

STATA Model Output:

```

Source |      SS      df      MS      Number of obs      =      550
-----+-----
Model | 18529.397      7    2647.05671    F(7, 542)      =      30.04
Residual | 47767.1412    542    88.1312568    Prob > F      =      0.0000
-----+-----
Total | 66296.5382    549    120.758722    R-squared      =      0.2795
                                           Adj R-squared  =      0.2702
                                           Root MSE      =      9.3878

cognition |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-----+-----
age85 | -.2322489   .1893912    -1.23   0.221    - .6042797   .1397818    Beta1
grip9 | .6968828   .2387998     2.92   0.004     .2277964   1.165969    Beta2
sexmw | -3.626534   .9111454    -3.98   0.000    -5.416343   -1.836725    Beta3
demnf | -5.748826   1.020154    -5.64   0.000    -7.752765   -3.744887    Beta4
demnc | -16.49476   1.523337   -10.83   0.000   -19.48713   -13.50239    Beta5
c.sexmw#c.age85 | -.2947856   .2439482    -1.21   0.227    - .7739853   .1844141    Beta6
c.sexmw#c.grip9 | -.1767274   .3066309    -0.58   0.565    - .7790578   .4256031    Beta7
_cons | 29.17207    .7583565    38.47   0.000     27.6824    30.66175    Beta0

.      lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
( 1)  - demnf + demnc = 0

cognition |      Coef.   Std. Err.      t    P>|t|     [95% Conf. Interval]
-----+-----
(1) | -10.74593   1.708752    -6.29   0.000   -14.10252   -7.389346

.      test (c.demnf=0) (c.demnc=0) // DFnum=2 F-test for Demgroup
( 1)  demnf = 0
( 2)  demnc = 0
      F( 2, 542) = 67.19
      Prob > F = 0.0000

```



```
. test (c.sexmw#c.age85=0) (c.sexmw#c.grip9=0) // DFnun=2 F-test for new interactions
( 1) c.sexmw#c.age85 = 0
( 2) c.sexmw#c.grip9 = 0
      F( 2, 542) = 0.81
      Prob > F = 0.4438
```

```
// Simple slopes of age by sex
. lincom c.age85*1 + c.sexmw#c.age85*0 // Age Slope for Men
( 1) age85 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	-.2322489	.1893912	-1.23	0.221	-.6042797	.1397818

```
. lincom c.age85*1 + c.sexmw#c.age85*1 // Age Slope for Women
( 1) age85 + c.sexmw#c.age85 = 0
```



cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	-.5270345	.1537664	-3.43	0.001	-.8290857	-.2249833

```
// Simple slopes of grip by sex
. lincom c.grip9*1 + c.sexmw#c.grip9*0 // Grip Slope for Men
( 1) grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	.6968828	.2387998	2.92	0.004	.2277964	1.165969

```
. lincom c.grip9*1 + c.sexmw#c.grip9*1 // Grip Slope for Women
( 1) grip9 + c.sexmw#c.grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	.5201555	.1930589	2.69	0.007	.1409201	.8993908



```
// Stata: calculate semipartial eta2 for 2 new interactions using SS from reduced model
display "STATA Reduced Model to Get SS for 2 interactions (not included)"
regress cognition c.age85 c.grip9 c.sexmw c.age85#c.grip9, level(95)
```

Source	SS	df	MS	Number of obs	=	550
Model	18385.9793	5	3677.19586	F(5, 544)	=	41.75
Residual	47910.5589	544	88.0708803	Prob > F	=	0.0000
Total	66296.5382	549	120.758722	R-squared	=	0.2773
				Adj R-squared	=	0.2707
				Root MSE	=	9.3846




```
// sr2 = (SSfull-SSreduced)/SStotal
display (18529.397-18385.9793)/66296.5382 // sr2 for sex*age, sex*grip = .00216328
```

SAS Syntax and Output for Plots:

```
* Labeling saved ESTIMATES for use in plot;
* INDEX finds value in parentheses for that column;
DATA work.EstSexInteract; LENGTH sex $6; SET work.EstSexInteract;
  IF INDEX(Parameter,"Men")>0 THEN sex="Men";
  IF INDEX(Parameter,"Women")>0 THEN sex="Women";
  IF INDEX(Parameter,"Age=80")>0 THEN age=80;
  IF INDEX(Parameter,"Age=90")>0 THEN age=90;
  IF INDEX(Parameter,"Grip=6")>0 THEN grip=6;
  IF INDEX(Parameter,"Grip=12")>0 THEN grip=12;
RUN;

* Plot ESTIMATES -- grip as X by sex holding age=80;
PROC SGPLOT DATA=work.EstSexInteract; WHERE age=80;
  SERIES x=grip y=Estimate / GROUP=sex;
  XAXIS GRID LABEL="Grip Strength" VALUES=(6 TO 12 BY 3);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5);
RUN; QUIT;

* Plot ESTIMATES -- age as X by sex holding grip=6;
PROC SGPLOT DATA=work.EstSexInteract; WHERE grip=6;
  SERIES x=age y=Estimate / GROUP=sex;
  XAXIS GRID LABEL="Years of Age" VALUES=(80 TO 90 BY 5);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5);
RUN; QUIT;
```



Example Results Section for Sex Interactions Model [notes about what also to include]:

Equation 1:

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{SexMW}_i)(\text{Age}_i - 85) + \beta_7(\text{SexMW}_i)(\text{Grip}_i - 9) + e_i$$

We estimated a general linear model (as shown in Equation 1) to examine the extent to which cognition could be predicted from linear slopes of age (centered such that 0 = 85 years) and grip strength (centered such that 0 = 9 pounds per square inch), sex (0 = men, 1 = women), and dementia status (none vs. future; none vs. current), as well as interactions of sex with age and sex with grip strength. Although the model accounted for a significant amount of variance in cognition, $F(7,542) = 30.04$, $\text{MSE} = 88.13$, $p < .0001$, $R^2 = .280$, the addition of the two interactions did not significantly improve prediction relative to the main effects model, $F(2,542) = 0.81$, $\text{MSE} = 88.13$, $p = .444$, change in $R^2 = .002$. Results indicated that the effects of age and grip strength did not differ significantly between men and women, and so these nonsignificant interactions were removed from the model.

NONE OF THIS IS NEEDED FOR HW6, but this model is presented in Hoffman (2015) ch. 2...

SAS Syntax: Removing Sex Interactions; Adding Interaction of Age by Grip Strength

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + e_i$$

```
TITLE1 "SAS Eq 2.9: GLM with Age by Grip Interaction";
TITLE2 "Using dataset with fake people to get predicted outcomes as saved variable";
* Estimate model on data with fake people to make predictions;
PROC GLM DATA=work.Example6 NAMELEN=100;
MODEL cognition = age85 grip9 sexMW demNF demNC age85*grip9
/ ALPHA=.05 CLPARM SOLUTION SS3 EFFECTSIZE;
ESTIMATE "Mean Diff: Fut vs. Cur" demNF -1 demNC 1; * B5-B4;
CONTRAST "DFnum=2 F-test for Demgroup" demNF 1, demNC 1;
CONTRAST "DFnum=3 F-test for age, grip, age*grip" age85 1, grip9 1, age85*grip9 1;
* Request columns of predicted outcome and SE for all cases;
OUTPUT OUT=work.PredOutcomes PREDICTED=Yhat STDP=SEyhat;
```

We can use the model equation to calculate the **simple age slope** at any *grip strength* (as the moderator):

$$\begin{aligned} \text{Simple Age Slope} &= \beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \\ &= [\beta_1 + \beta_6(\text{Grip}_i - 9)] \text{ that multiplies } (\text{Age}_i - 85) \end{aligned}$$

```
ESTIMATE "Age Slope at Grip = 6" age85 1 age85*grip9 -3;
ESTIMATE "Age Slope at Grip = 9" age85 1 age85*grip9 0;
ESTIMATE "Age Slope at Grip = 12" age85 1 age85*grip9 3;
```

We can also use the model equation to calculate the **simple grip strength slope** at any *age* (as the moderator):

$$\begin{aligned} \text{Simple Grip Slope} &= \beta_2(\text{Grip}_i - 9) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \\ &= [\beta_2 + \beta_6(\text{Age}_i - 85)] \text{ that multiplies } (\text{Grip}_i - 9) \end{aligned}$$

```
ESTIMATE "Grip Slope at Age = 80" grip9 1 age85*grip9 -5;
ESTIMATE "Grip Slope at Age = 85" grip9 1 age85*grip9 0;
ESTIMATE "Grip Slope at Age = 90" grip9 1 age85*grip9 5;
```

If you are using “fake people” then you do NOT need to write these ESTIMATE statements also.

```
* Pred cognition outcomes holding sexMW=men, demNF=none, and demNC=none;
ESTIMATE "Yhat for Age=80 Grip=6" intercept 1 age85 -5 grip9 -3 age85*grip9 15;
ESTIMATE "Yhat for Age=80 Grip=9" intercept 1 age85 -5 grip9 0 age85*grip9 0;
ESTIMATE "Yhat for Age=80 Grip=12" intercept 1 age85 -5 grip9 3 age85*grip9 -15;
ESTIMATE "Yhat for Age=85 Grip=6" intercept 1 age85 0 grip9 -3 age85*grip9 0;
ESTIMATE "Yhat for Age=85 Grip=9" intercept 1 age85 0 grip9 0 age85*grip9 0;
ESTIMATE "Yhat for Age=85 Grip=12" intercept 1 age85 0 grip9 3 age85*grip9 0;
ESTIMATE "Yhat for Age=90 Grip=6" intercept 1 age85 5 grip9 -3 age85*grip9 -15;
ESTIMATE "Yhat for Age=90 Grip=9" intercept 1 age85 5 grip9 0 age85*grip9 0;
ESTIMATE "Yhat for Age=90 Grip=12" intercept 1 age85 5 grip9 3 age85*grip9 15;
ODS OUTPUT Estimates=EstAgebyGrip; * Save estimates to dataset;
RUN; TITLE1; TITLE2;
```

STATA Syntax: Removing Sex Interactions; Adding Interaction of Age by Grip Strength

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + e_i$$

```
display "STATA Eq 2.9: Adding Age by Grip Interaction"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc c.age85#c.grip9, level(95)
lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
test (c.demnf=0) (c.demnc=0) // DFnum=2 F-test for Demgroup
test (c.age85=0) (c.grip9=0) (c.age85#c.grip9=0) // DFnum=3 F-test for age, grip, age*grip
```

We can use the model equation to calculate the **simple age slope** at any *grip strength* (as the moderator):

$$\begin{aligned}\text{Simple Age Slope} &= \beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \\ &= [\beta_1 + \beta_6(\text{Grip}_i - 9)] \text{ that multiplies } (\text{Age}_i - 85)\end{aligned}$$

```
// dydx in margins provides simple slopes for that variable by (from(by)to) moderator
margins, at(c.grip9=(-3(3)3)) dydx(c.age85) vsquish // Age Slope per Grip
lincom c.age85*1 + c.age85#c.grip9*-3 // Age Slope at Grip = 6
lincom c.age85*1 + c.age85#c.grip9*0 // Age Slope at Grip = 9
lincom c.age85*1 + c.age85#c.grip9*3 // Age Slope at Grip = 12
```

We can also use the model equation to calculate the **simple grip strength slope** at any *age* (as the moderator):

$$\begin{aligned}\text{Simple Grip Slope} &= \beta_2(\text{Grip}_i - 9) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \\ &= [\beta_2 + \beta_6(\text{Age}_i - 85)] \text{ that multiplies } (\text{Grip}_i - 9)\end{aligned}$$

```
// dydx in margins provides simple slopes for that variable by (from(by)to) moderator
margins, at(c.age85=(-5(5)5)) dydx(c.grip9) vsquish // Grip per Age
lincom c.grip9*1 + c.age85#c.grip9*-5 // Grip Slope at Age = 80
lincom c.grip9*1 + c.age85#c.grip9*0 // Grip Slope at Age = 85
lincom c.grip9*1 + c.age85#c.grip9*5 // Grip Slope at Age = 90

// Get predicted outcomes for each combination of (from(by)to)
margins, at(c.age85=(-5(5)5) c.grip9=(-3(3)3) c.sexmw=0 c.demnf=0 c.demnc=0) vsquish
marginsplot, xdimension(age85) // Plot pred outcomes by age
marginsplot, xdimension(grip9) // Plot pred outcomes by grip
```

SAS Model Output:

SAS Eq 2.9: GLM with Age by Grip Interaction

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	6	19185.04106	3197.50684	36.85	<.0001
Error	543	47111.49712	86.76150		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.289382	37.52580	9.314586	24.82182

Source	Total Variation Accounted For				
	Semipartial		Conservative		
	Semipartial Eta-Square	Omega- Square	95% Confidence Limits		
age85	0.0101	0.0088	0.0002	0.0329	ok but conditional
grip9	0.0227	0.0214	0.0046	0.0528	ok but conditional
sexMW	0.0199	0.0185	0.0034	0.0486	ok, unconditional
demNF	0.0447	0.0433	0.0169	0.0827	not ok, needs combined
demNC	0.1520	0.1505	0.1010	0.2062	not ok, needs combined
age85*grip9	0.0121	0.0107	0.0007	0.0363	ok, unconditional
DFnum=2 F-test for Demgroup	0.1772	0.1744	0.1218	0.2315	ok, from CONTRAST
DFnum=3 F-test for age, grip, age*grip	0.0573	0.0533	0.0224	0.0950	ok, from CONTRAST
Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
DFnum=2 F-test for Demgroup	2	11747.60589	5873.80294	67.70	<.0001
DFnum=3 F-test for age, grip, age*grip	3	3799.85696	1266.61899	14.60	<.0001

Table of Model-Estimated Fixed Effects (normally is last)

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits		
Intercept	29.40780315	0.69490615	42.32	<.0001	28.04276953	30.77283677	Beta0
age85	-0.33396058	0.12035656	-2.77	0.0057	-0.57038207	-0.09753908	Beta1
grip9	0.61941863	0.14874241	4.16	<.0001	0.32723761	0.91159964	Beta2
sexMW	-3.45563720	0.88727488	-3.89	0.0001	-5.19854887	-1.71272552	Beta3
demNF	-5.92254309	1.01363159	-5.84	<.0001	-7.91366261	-3.93142358	Beta4
demNC	-16.30040485	1.51254730	-10.78	<.0001	-19.27156564	-13.32924405	Beta5
age85*grip9	0.12301848	0.04053626	3.03	0.0025	0.04339138	0.20264558	Beta6

Interpret these fixed effects:Simple main effect of Age β_1 =Simple main effect of Grip Strength β_2 =Interpret Age by Grip Strength $\beta_6 \rightarrow$ Age as Simple Slope, Grip as Moderator:Interpret Age by Grip Strength $\beta_6 \rightarrow$ Grip as Simple Slope, Age as Moderator:**Table of Extra Requested Linear Combinations of Model-Estimated Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr > t	95% Confidence Limits	
Mean Diff: Fut vs. Cur	-10.3778618	1.69983087	-6.11	<.0001	-13.7169116	-7.0388119
Age Slope at Grip = 6	-0.7030160	0.15336958	-4.58	<.0001	-1.0042864	-0.4017456
Age Slope at Grip = 9	-0.3339606	0.12035656	-2.77	0.0057	-0.5703821	-0.0975391
Age Slope at Grip = 12	0.0350949	0.18715387	0.19	0.8513	-0.3325394	0.4027291
Grip Slope at Age = 80	0.0043262	0.24733508	0.02	0.9861	-0.4815246	0.4901770
Grip Slope at Age = 85	0.6194186	0.14874241	4.16	<.0001	0.3272376	0.9115996
Grip Slope at Age = 90	1.2345110	0.25540829	4.83	<.0001	0.7328017	1.7362204
Yhat for Age=80 Grip=6	31.0646273	1.26047290	24.65	<.0001	28.5886270	33.5406277
Yhat for Age=80 Grip=9	31.0776060	0.91678862	33.90	<.0001	29.2767193	32.8784928
Yhat for Age=80 Grip=12	31.0905847	1.09240761	28.46	<.0001	28.9447221	33.2364473
Yhat for Age=85 Grip=6	27.5495473	0.93087754	29.60	<.0001	25.7209850	29.3781095
Yhat for Age=85 Grip=9	29.4078031	0.69490615	42.32	<.0001	28.0427695	30.7728368
Yhat for Age=85 Grip=12	31.2660590	0.70533225	44.33	<.0001	29.8805450	32.6515731
Yhat for Age=90 Grip=6	24.0344672	1.14908030	20.92	<.0001	21.7772801	26.2916544
Yhat for Age=90 Grip=9	27.7380003	0.92172276	30.09	<.0001	25.9274212	29.5485794
Yhat for Age=90 Grip=12	31.4415333	1.24617867	25.23	<.0001	28.9936117	33.8894549

STATA Model Output:

Source	SS	df	MS	Number of obs	=	550
Model	19185.0411	6	3197.50684	F(6, 543)	=	36.85
Residual	47111.4971	543	86.7615048	Prob > F	=	0.0000
Total	66296.5382	549	120.758722	R-squared	=	0.2894
				Adj R-squared	=	0.2815
				Root MSE	=	9.3146

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]		
age85	-.3339606	.1203566	-2.77	0.006	-.5703821	-.0975391	Beta1
grip9	.6194186	.1487424	4.16	0.000	.3272376	.9115996	Beta2
sexmw	-3.455637	.8872749	-3.89	0.000	-5.198549	-1.712726	Beta3
demnf	-5.922543	1.013632	-5.84	0.000	-7.913663	-3.931424	Beta4
demnc	-16.3004	1.512547	-10.78	0.000	-19.27157	-13.32924	Beta5
c.age85#c.grip9	.1230185	.0405363	3.03	0.003	.0433914	.2026456	Beta6
_cons	29.4078	.6949062	42.32	0.000	28.04277	30.77284	Beta0

```
. lincom c.demnf*-1 + c.demnc*1 // Mean Diff: Future vs. Current = B5-B4
(1) - demnf + demnc = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	-10.37786	1.699831	-6.11	0.000	-13.71691	-7.038812

```
. test (c.demnf=0)(c.demnc=0) // DFnun=2 F-test for Demgroup
```

```
(1) demnf = 0
(2) demnc = 0
F( 2, 543) = 67.70
Prob > F = 0.0000
```

```
. test (c.age85=0)(c.grip9=0) (c.age85#c.grip9=0) // DFnun=3 F-test for age, grip, age*grip
```

```
(1) age85 = 0
(2) grip9 = 0
(3) c.age85#c.grip9 = 0
F( 3, 543) = 14.60
Prob > F = 0.0000
```

Expression : Linear prediction, predict()

dy/dx w.r.t. : age85

1._at : grip9 = -3

2._at : grip9 = 0

3._at : grip9 = 3

		Delta-method				
		dy/dx	Std. Err.	t	P> t	[95% Conf. Interval]
age85	_at					
	1	-.703016	.1533696	-4.58	0.000	-1.004286 -.4017456
	2	-.3339606	.1203566	-2.77	0.006	-.5703821 -.0975391
	3	.0350949	.1871539	0.19	0.851	-.3325394 .4027291

```
. lincom c.age85*1 + c.age85#c.grip9*-3 // Age Slope at Grip = 6
```

```
(1) age85 - 3*c.age85#c.grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	-.703016	.1533696	-4.58	0.000	-1.004286	-.4017456

```
. lincom c.age85*1 + c.age85#c.grip9*0 // Age Slope at Grip = 9
```

```
(1) age85 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	-.3339606	.1203566	-2.77	0.006	-.5703821	-.0975391

```
. lincom c.age85*1 + c.age85#c.grip9*3 // Age Slope at Grip = 12
```

```
(1) age85 + 3*c.age85#c.grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
(1)	.0350949	.1871539	0.19	0.851	-.3325394	.4027291

```
. // dydx in margins provides simple slopes for that variable by (from(by)to) moderator
```

```
. margins, at(c.age85=(-5(5)5)) dydx(c.grip9) vsquish // Grip Slope per Age
```

Expression : Linear prediction, predict()

dy/dx w.r.t. : grip9

1._at : age85 = -5

2._at : age85 = 0

3._at : age85 = 5

		Delta-method				
		dy/dx	Std. Err.	t	P> t	[95% Conf. Interval]
grip9	_at					
	1	.0043262	.2473351	0.02	0.986	-.4815246 .490177
	2	.6194186	.1487424	4.16	0.000	.3272376 .9115996

```
3 | 1.234511 .2554083 4.83 0.000 .7328017 1.73622
```

```
. lincom c.grip9*1 + c.age85#c.grip9*-5 // Grip Slope at Age = 80
(1) grip9 - 5*c.age85#c.grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
(1)	.0043262	.2473351	0.02	0.986	-.4815246 .490177

```
. lincom c.grip9*1 + c.age85#c.grip9*0 // Grip Slope at Age = 85
(1) grip9 = 0
```



cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
(1)	.6194186	.1487424	4.16	0.000	.3272376 .9115996

```
. lincom c.grip9*1 + c.age85#c.grip9*5 // Grip Slope at Age = 90
(1) grip9 + 5*c.age85#c.grip9 = 0
```

cognition	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
(1)	1.234511	.2554083	4.83	0.000	.7328017 1.73622

_at	Delta-method		t	P> t	[95% Conf. Interval]	
	Margin	Std. Err.				
1	31.06463	1.260473	24.65	0.000	28.58863	33.54063
2	31.07761	.9167886	33.90	0.000	29.27672	32.87849
3	31.09058	1.092408	28.46	0.000	28.94472	33.23645
4	27.54955	.9308775	29.60	0.000	25.72099	29.37811
5	29.4078	.6949062	42.32	0.000	28.04277	30.77284
6	31.26606	.7053323	44.33	0.000	29.88054	32.65157
7	24.03447	1.14908	20.92	0.000	21.77728	26.29165
8	27.738	.9217228	30.09	0.000	25.92742	29.54858
9	31.44153	1.246179	25.23	0.000	28.99361	33.88945

Although annoying that they are not labeled here, a long table preceded this MARGINS result that says what the predictor values are for each of these 9 predicted outcomes.



STATA Syntax to get same semipartial η^2 values given by CONTRAST in SAS

```
display "STATA Eq 2.9: Full Model with Age by Grip Interaction (repeated)"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc c.age85#c.grip9, level(95)
```

Source	SS	df	MS	Number of obs	=	550
Model	19185.0411	6	3197.50684	F(6, 543)	=	36.85
Residual	47111.4971	543	86.7615048	Prob > F	=	0.0000
				R-squared	=	0.2894
				Adj R-squared	=	0.2815
Total	66296.5382	549	120.758722	Root MSE	=	9.3146

```
// Get semipartial eta2 for single terms
pcorr cognition age85 grip9 sexmw demnf demnc agegrip
```

Variable	Partial Corr.	Semipartial Corr.	Partial Corr.^2	Semipartial Corr.^2	Significance Value
age85	-0.1182	-0.1004	0.0140	0.0101	0.0057
grip9	0.1759	0.1506	0.0309	0.0227	0.0000
sexmw	-0.1648	-0.1409	0.0272	0.0199	0.0001
demnf	-0.2432	-0.2114	0.0592	0.0447	0.0000
demnc	-0.4198	-0.3899	0.1762	0.1520	0.0000
agegrip	0.1291	0.1098	0.0167	0.0121	0.0025

```
// Stata: calculate semipartial eta2 for demgroup using SS from reduced model
display "STATA Reduced Model to Get SS for demnf and demnc (not included)"
regress cognition c.age85 c.grip9 c.sexmw c.age85#c.grip9, level(95)
```

Source	SS	df	MS	Number of obs	=	550
Model	7437.43517	4	1859.35879	F(4, 545)	=	17.22
Residual	58859.103	545	107.998354	Prob > F	=	0.0000
				R-squared	=	0.1122
				Adj R-squared	=	0.1057
Total	66296.5382	549	120.758722	Root MSE	=	10.392

```
// sr2 = (SSfull-SSreduced)/SStotal
display (19185.0411-7437.43517)/66296.5382 // sr2 for demgroup = .17719788
```

```
// Stata: calculate semipartial eta2 for age, grip, age*agrip using SS from reduced model
display "STATA Reduced Model to Get SS for age, grip, age*agrip (not included)"
regress cognition c.sexmw c.demnf c.demnc, level(95)
```



Source	SS	df	MS	Number of obs	=	550
Model	15385.1841	3	5128.3947	F(3, 546)	=	55.00
Residual	50911.3541	546	93.2442382	Prob > F	=	0.0000
				R-squared	=	0.2321
				Adj R-squared	=	0.2278
Total	66296.5382	549	120.758722	Root MSE	=	9.6563

```
// sr2 = (SSfull-SSreduced)/SStotal
display (19185.0411-15385.1841)/66296.5382 // sr2 for age,grip,age*grip = .05731607
```


SAS Syntax and Output for Plots:

```
* Plot saved predicted values for fake people -- age as X;
PROC SGPLOT DATA=work.PredOutcomes; WHERE PersonID=-99; * Only for fake people;
  SERIES x=age y=Yhat / GROUP=grip;
  XAXIS GRID LABEL="Years of Age" VALUES=(80 TO 90 BY 5);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5); RUN; QUIT;

* Plot saved predicted values for fake people -- grip as X;
PROC SGPLOT DATA=work.PredOutcomes; WHERE PersonID=-99; * Only for fake people;
  SERIES x=grip y=Yhat / GROUP=age;
  XAXIS GRID LABEL="Grip Strength" VALUES=(6 TO 12 BY 3);
  YAXIS GRID LABEL="Predicted Cognition" VALUES=(15 TO 45 BY 5); RUN; QUIT;
```



SAS Syntax and Output for Regions of Significance:

To get all the necessary info to calculate regions of significance for the age*grip interaction, we need to change to PROC MIXED, which does GLMs but has many other options, including COVB for the asymptotic covariance matrix of the fixed effects, in which the diagonal is their squared standard errors, and the off-diagonals give the covariances among their SEs. COVB is provided in STATA as “estat vce” after estimating a model with fixed effects (as given in the STATA syntax below).

```
TITLE1 "SAS Eq 2.9: MIXED with Age by Grip Interaction to Get COVB";
TITLE2 "Using dataset with fake people to get predicted outcomes as saved variable";
* Estimate model on data with fake people to make predictions;
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 METHOD=REML;
  MODEL cognition = age85 grip9 sexMW demNF demNC age85*grip9
    / SOLUTION DDFM=BW COVB OUTPM=PredOutcomes;
  * Saving info for regions to datasets: fixed effects and COVB;
  ODS OUTPUT SolutionF=FixAgeGrip COVB=CovBAgeGrip;
  RUN; TITLE1; TITLE2;

* Get correlation of yhat with y to compute R2 (not provided);
PROC CORR NOSIMPLE DATA=work.PredOutcomes; VAR cognition; WITH pred; RUN;

display "STATA Eq 2.9: GLM with Age by Grip Interaction adding VCE for regions"
regress cognition c.age85 c.grip9 c.sexmw c.demnf c.demnc c.age85#c.grip9, level(95)
estat vce // Asymptotic covariance matrix of fixed effects for regions
```

From COVB using SAS MIXED instead of GLM (bolded values needed for %regions macro):

Covariance Matrix for Fixed Effects								
Row	Effect	Col1	Col2	Col3	Col4	Col5	Col6	Col7
1	Intercept	0.4829	0.000454	-0.03075	-0.4507	-0.1820	-0.2263	0.001916
2	age85	0.000454	0.01449	0.003317	0.005024	-0.00413	-0.00115	0.000959
3	grip9	-0.03075	0.003317	0.02212	0.05374	-0.01339	-0.00030	0.000203
4	sexMW	-0.4507	0.005024	0.05374	0.7873	-0.07102	0.02371	0.002695
5	demNF	-0.1820	-0.00413	-0.01339	-0.07102	1.0274	0.2129	-0.00268
6	demNC	-0.2263	-0.00115	-0.00030	0.02371	0.2129	2.2878	0.002396
7	age85*grip9	0.001916	0.000959	0.000203	0.002695	-0.00268	0.002396	0.001643

See excel sheet for calculations, as provided by the SAS macro %Regions below:

* Call SAS macro for regions of significance for main effects of interaction;
 %Regions(FixData=FixAgeGrip, CovBData=CovBAgeGrip, Pred=grip9, Mod=age85,
 ModCenter=85, Interact=age85*grip9, Order=6);

Regions of significance for age85*grip9 interaction:

The grip9 slope will be significant at centered values of age85 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered	Centered	Uncentered	Uncentered
Lower	Upper	Lower	Upper
-14.8174	-2.28519	70.1826	82.7148

So the grip strength slope will be significantly negative below age = 70.19 years, nonsignificant between age = 70.19 and 82.71 years, and significantly positive after age = 82.71 years.

* Call SAS macro for regions of significance for main effects of interaction;
 %Regions(FixData=FixAgeGrip, CovBData=CovBAgeGrip, Pred=age85, Mod=grip9,
 ModCenter=9, Interact=age85*grip9, Order=6)

Regions of significance for age85*grip9 interaction:

The age85 slope will be significant at centered values of grip9 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered	Centered	Uncentered	Uncentered
Lower	Upper	Lower	Upper
0.66541	9.52041	9.66541	18.5204

So the age slope will be significantly negative below grip = 9.67 pounds, nonsignificant between grip = 9.67 and 18.52 pounds, and significantly positive after grip = 18.52 pounds.

Example Results Section for Age*Grip Model Using SAS Output [notes about what also to include]:

We estimated a general linear model (as shown in Equation 2) to examine the extent to which cognition could be predicted from linear slopes of age (centered such that 0 = 85 years), grip strength (centered such that 0 = 9 pounds per square inch), and their interaction, as well as sex (0 = men, 1 = women), and dementia status (none vs. future; none vs. current). The model accounted for a significant amount of variance in cognition, $F(6,543) = 36.85$, $MSE = 86.67$, $p < .0001$, $R^2 = .289$. Table 2 provides the model results, including the fixed effects estimated directly in the model, as well as their linear combinations in order to provide simple slopes by which to describe the age by grip strength interaction. Effect sizes are given below using semipartial eta-squared (η^2) values, the contribution of each slopes or combinations of slopes to the total model R^2 .

Equation 2:

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + e_i$$

Table 1: Model Results (bold values indicate $p < .0001$)

	Model Parameter	Est	SE	$p <$	η^2
β_0	Intercept	29.41	0.69	.001	
	Age Slope				
	Grip Strength (0 = 6 lbs)	-0.70	0.15	.001	
β_1	Grip Strength (0 = 9 lbs)	-0.33	0.12	.006	.010
	Grip Strength (0 = 12 lbs)	0.04	0.19	.851	
	Grip Strength Slope				
	Age (0 = 80 years)	0.00	0.25	.986	
β_2	Age (0 = 85 years)	0.62	0.15	.001	.023
	Age (0 = 90 years)	1.23	0.26	.001	
β_3	Sex (0 = Men, 1 = Women)	-3.46	0.89	.001	.020
	Dementia Group:				
β_4	None vs. Future	-5.92	1.01	.001	
β_5	None vs. Current	-16.30	1.51	.001	
$\beta_5 - \beta_4$	Future vs. Current	-10.38	1.70	.001	
β_6	Age by Grip Interaction	0.12	0.04	.003	.012

Note: η^2 = semipartial eta-squared for amount of model R^2 due to that predictor.

The intercept $\beta_0 = 29.41$ is the expected cognition outcome for an 85-year-old man with 9 pounds of grip strength who will not be diagnosed with dementia later in the study. Women were predicted to have significantly lower cognition by the main effect of sex, $\beta_3 = -3.46$. Likewise, relative to the no dementia group, cognition was predicted to be significantly lower in the future group by $\beta_4 = -5.92$ and in the current group by $\beta_5 = -16.30$. The future and current groups also differed significantly by $\beta_5 - \beta_4 = -10.38$ and the omnibus effect for differences across the three groups was significant as well, $F(2,543) = 67.70$, $p < .0001$, $\eta^2 = .177$.

The simple main effect of age $\beta_1 = 0.33$ indicated that cognition is predicted to be significantly lower by 0.33 for every additional year of age (in persons with grip strength of 9 pounds). The simple main effect of grip strength $\beta_2 = 0.62$ indicated that cognition is predicted to be significantly greater by 0.62 for every additional pound of grip strength (in persons who are age 85). As shown in Figure 2.1, the age by grip strength interaction $\beta_6 = 0.12$ indicated the age slope predicting cognition became significantly less negative by 0.12 for each additional pound of grip strength (as shown by the differences in slopes of the lines). Equivalently, the grip strength slope predicting cognition became significantly more positive by 0.12 for each additional year of age (as shown by the differences in the vertical distance between the lines). The combination of the linear slopes for age, grip strength, and their interaction accounted for significant variance, $F(3,543) = 14.60$, $p < .0001$, $\eta^2 = .057$.

Simple slopes for age and grip at other levels of the interacting predictor are also given in Table 1. To further describe the age by grip strength interaction, the regions along each moderator through which the other main effect is expected to be significant were then calculated using the fixed effect estimates and their associated covariance matrix, as described in Hoffman (2015). For the effect of age, the obtained threshold values of grip strength were 9.67 and 18.52 pounds. Given the range of grip strength of 0 to 19 pounds in the current sample ($M \approx 9$), the effect of age is expected to be negative for about half of the sample (below 9.67 pounds), the effect of age is expected to be nonsignificant for the other half (between 9.67 and 18.52 pounds), and the effect of age is expected to be positive for almost no one (above 18.52 pounds). Similarly, for the effect of grip strength, the obtained threshold values of age were 70.18 and 82.71 years. Given the range of age of 80 to 97 years in the sample ($M \approx 85$), the effect of grip strength is expected to be negative for no one (below 70.18 years), the effect of grip strength is expected to be nonsignificant for a small part of the sample (between 70.18 and 82.71 years), and the effect of grip strength is expected to be positive for the majority of the sample (above 82.71 years).